Logistická křivka

Při modelování saturace trhu nějakým výrobkem je často vhodné užít tzv. S-křivek. Dvěma nejznámnějšími jsou Gompertzova a logistická. Podíváme se na ně z analytického hlediska a srovnáme jejich průběh.

Logistická křivka má tento tvar:

```
> f9(t):= c/(1+a*b^t);
```

$$f9(t) := \frac{c}{1 + a\, b^t}$$

kde c>0, a>1, 0<b<1. Z toho je zřejmé, že pro t jdoucí k nekonečnu se f9 zdola blíží ke konstantě c.Ověřím to výpočtem:

```
> assume(c>0);
  assume(a>1);
  assume(b, RealRange(Open(0),Open(1)));
  limit(f9(t),t=infinity);
```

$$c$$

Odbočka:

Na příkaz assume je nutné dát pozor. Svědčí o tom následující:

```
> assume(b, RealRange(0,1));
  limit(b^t, t=infinity);
```

$$0$$

Situace b=1, kdy limita vyjde 1 je MAPLEm ignorována !

Limita zprava v t=0 je

```
> limit(f9(t), t=0, right);
```

$$\frac{c}{1 + a}$$

Zakreslím průběh funkce pro několik parametrů a a b (u c to nemá cenu, neboť jen vertikálně "roztahuje" graf).

```
> ff9:=(t,a,b,c)-> c/(1+a*b^t);
```

$$ff9 := (t, a, b, c) \rightarrow \frac{c}{1 + a\, b^t}$$

```
> kriv9:=[seq(ff9(t,i,0.5,5),i=[2,10,1000,1000])]:
  plot(kriv9, t=0..15,color=black,
  thickness=[2,0,0,0],title=`Parametr a=2,10,100,1000 (a=2
  tučně, další doprava)`);
>
```




```
> kriv9a:=[seq(ff9(t,10,i,5),i=[0.01, 0.5, 0.7, 0.99])]:
  plot(kriv9a, t=0..15,color=black,
  thickness=[2,0,0,0],title=`Parametr b=0.01, 0.5, 0.7, 0.99
  (b=0.01 tučně, pak doprava)`);
```



Derivace (na obrázku znázorněna tučně):

```
> diff(f9(t),t);
  fa9(t):= 3/(1+10*0.5^t):
  plot( [diff(fa9(t),t),fa9(t)], t=0..10, color=black,
  thickness=[2,0]);
```

$$-\frac{c\,a\,b^t \ln(b\sim)}{(1+a\,b^t)^2}$$



Druhá derivace je :

```
> diff(f9(t),t$2);
```

$$\frac{2\,c\,a^2\,(b^t)^2\,\ln(b\sim)^2}{(1+a\,b^t)^3}-\frac{c\,a\,b^t\,\ln(b\sim)^2}{(1+a\,b^t)^2}$$

Inflexe tedy v:

```
> solve(diff(f9(t),t$2)=0,t);
```

$$-\frac{\ln(a)}{\ln(b)}$$

Oblast konvexity: MAPLE nerovnici nevyřeší:

```
> solve(diff(f9(t),t$2)>0,t);
```

$$\{t<-\frac{\ln(a\sim)}{\ln(b\sim)}\}$$

Jelikož druhá derivace existuje pro všechna t a nulová je v jediném bodě ($-\frac{\ln(a)}{\ln(b)}$) stačí dosadit. Nejprve ($-\frac{2\ln(a)}{\ln(b)}$):

```
> d2_9(t):=diff(f9(t),t$2):
  subs( t=-2*ln(a)/ln(b),d2_9(t));
```

$$\frac{2\, c\, a^2 \left(b^{\left(-\frac{2\ln(a\sim)}{\ln(b\sim)}\right)}\right)^2 \ln(b\sim)^2}{\left(1+a\, b^{\left(-\frac{2\ln(a\sim)}{\ln(b\sim)}\right)}\right)^3} - \frac{c\, a\, b^{\left(-\frac{2\ln(a\sim)}{\ln(b\sim)}\right)} \ln(b\sim)^2}{\left(1+a\, b^{\left(-\frac{2\ln(a\sim)}{\ln(b\sim)}\right)}\right)^2}$$

```
> simplify(%);
>
```

$$-\frac{c\sim\, a\sim \ln(b\sim)^2\, (a\sim - 1)}{(1+a\sim)^3}$$

Což je nutně záporné, vzhledem k předpokladům o a,b,c.

```
> subs( t=-1/2*ln(a)/ln(b),d2_9(t));
```

$$\frac{2\, c\, a^2 \left(b^{\left(-1/2\frac{\ln(a\sim)}{\ln(b\sim)}\right)}\right)^2 \ln(b\sim)^2}{\left(1+a\, b^{\left(-1/2\frac{\ln(a\sim)}{\ln(b\sim)}\right)}\right)^3} - \frac{c\, a\, b^{\left(-1/2\frac{\ln(a\sim)}{\ln(b\sim)}\right)} \ln(b\sim)^2}{\left(1+a\, b^{\left(-1/2\frac{\ln(a\sim)}{\ln(b\sim)}\right)}\right)^2}$$

```
> simplify(%);
>
```

$$-\frac{c\sim \ln(b\sim)^2\, (-a\sim + \sqrt{a\sim}\,)}{(1+\sqrt{a\sim}\,)^3}$$

Což je kladné. Tedy od nuly do $(-\frac{\ln(a)}{\ln(b)})$ je logistická křivka konvexní, pak konkávní až do nekonečna. Inflexe je v $(-\frac{\ln(a)}{\ln(b)})$.

Významnou vlastností logistické křivky je středová symetrie podle inflexního bodu. Dokážu to:

```
> subs(t=t-ln(a)/ln(b),f9(t));
```

$$\frac{c}{1+a\, b^{\left(t-\frac{\ln(a\sim)}{\ln(b\sim)}\right)}}$$

```
> simplify(%);
>
```

$$\frac{c\sim}{1+b\sim^t}$$

```
> %-subs(t=0,%);
```

$$\frac{c\sim}{1+b\sim^t} - \frac{c\sim}{2}$$

```
> simplify(%);
>
```

$$-\frac{(-1+b\sim^t)\, c\sim}{2\,(1+b\sim^t)}$$

A to je lichá funkce:

```
> subs( t=-t, %);
>
```

$$-\frac{(-1+b\sim^{(-t)})\,c\sim}{2\,(1+b\sim^{(-t)})}$$

```
> simplify(%);
>
```

$$\frac{(-1+b\sim^{t})\,c\sim}{2\,(1+b\sim^{t})}$$

```
>
```